

A brief guide to Android Gradle

GDG Korea Android

Gradle?

Gradle combines the power and flexibility of Ant with the dependency management and conventions of Maven into a more effective way to build.

Build Tool

- Configuration (Static) + Build (Dynamic)
- Ant
 - Focus on Build
 - hard to configure
- Maven
 - Focus on Configuration
 - hard to costumize
- Alt.
 - Gradle - Groovy language (Dynamic), DSL (Static)
 - Apache Ivy

Groovy

- Runs on Java VM
- The principle of least surprise (JAVA)
- Built-in List, Range, Map, Closure
- Improved Loops, Switches
- Built-in Regex, Improved Strings
- Supports Dynamic typing and Static typing.

Groovy

```
class Greet {  
    def name  
    Greet(who) { name = who[0].toUpperCase() + who[1..-1] }  
    def salute() { println "Hello $name!" }  
}
```

```
g = new Greet('world')  
g.salute()
```

Task: Hello World

```
task hello {  
    doLast {  
        println 'Hello world!'  
    }  
}
```

gradle -q hello

Hello world!

Short version

```
task hello << {  
    println 'Hello world!'  
}
```


Other task example

```
task upper << {  
    String someString = 'mY_nAmE'  
    println "Original: " + someString  
    println "Upper case: " +  
someString.toUpperCase()  
}
```

Loop example

```
task count << {  
  4.times { print "$it " }  
}
```

Dependencies

```
task hello << {  
    println 'Hello world!'  
}
```

```
task intro(dependsOn: hello) << {  
    println "I'm Gradle"  
}
```

Dynamic tasks

```
4.times { counter ->  
    task "task$counter" << {  
        println "I'm task number $counter"  
    }  
}
```

```
> gradle -q task1
```

```
I'm task number 1
```

Manipulating existing tasks

```
4.times { counter ->
  task "task$counter" << {
    println "I'm task number $counter"
  }
}
task0.dependsOn task2, task3
> gradle -q task0
I'm task number 2
I'm task number 3
I'm task number 0
```

Default Tasks

```
defaultTasks 'clean', 'run'
```

```
task clean << {  
    println 'Default Cleaning!'  
}
```

```
task run << {  
    println 'Default Running!'  
}
```

```
task other << {  
    println "I'm not a default task!"  
}
```

Java Plugin

apply plugin: 'java'

gradle build

:compileJava

:processResources

:classes

:jar

:assemble

:compileTestJava

:processTestResources

:testClasses

:test

:check

:build

Adding repositories

```
repositories {  
    mavenCentral()  
}
```


Adding dependencies

```
dependencies {  
    compile group: 'commons-collections', name:  
'commons-collections', version: '3.2'  
    testCompile group: 'junit', name: 'junit', version:  
'4.+'  
}
```

Adding source sets

```
sourceSets {  
    main {  
        java {  
            srcDir 'src/java'  
        }  
        resources {  
            srcDir 'src/resources'  
        }  
    }  
}
```

More simple way

```
sourceSets {  
    main.java.srcDirs = ['src/java']  
    main.resources.srcDirs =  
    ['src/resources']  
}
```

Andorid build.gradle sample

```
buildscript {  
    repositories {  
        mavenCentral()  
    }  
  
    dependencies {  
        classpath 'com.android.tools.build:gradle:0.9.0'  
    }  
}  
  
apply plugin: 'android'  
  
android {  
    compileSdkVersion 19  
    buildToolsVersion "19.0.0"  
}
```

Android source sets

```
android {  
    sourceSets {  
        main {  
            manifest.srcFile 'AndroidManifest.xml'  
            java.srcDirs = ['src']  
            resources.srcDirs = ['src']  
            aidl.srcDirs = ['src']  
            renderscript.srcDirs = ['src']  
            res.srcDirs = ['res']  
            assets.srcDirs = ['assets']  
        }  
  
        androidTest.setRoot('tests')  
    }  
}
```

Android version configuration

```
android {  
    compileSdkVersion 19  
    buildToolsVersion "19.0.0"  
  
    defaultConfig {  
        versionCode 12  
        versionName "2.0"  
        minSdkVersion 16  
        targetSdkVersion 16  
    }  
}
```

Dynamic version

```
def computeVersionName() {  
    ...  
}  
  
android {  
    compileSdkVersion 19  
    buildToolsVersion "19.0.0"  
  
    defaultConfig {  
        versionCode 12  
        versionName computeVersionName()  
        minSdkVersion 16  
        targetSdkVersion 16  
    }  
}
```

More...

- <http://tools.android.com/tech-docs/new-build-system/user-guide>
- <http://www.vogella.com/tutorials/AndroidBuild/article.html>
- <http://kwonnam.pe.kr/wiki/gradle>